A MAPLE APPLICATION FOR TESTING SELF-ADJOINTNESS
ON QUANTUM GRAPHS
HELENE DALLMANN AND STEVEN COULTER

Abstract. In this paper we consider linear ordinary elliptic differential operators with smooth coefficients on finite quantum graphs. We discuss criteria
for the operator to be self-adjoint. This involves conditions on matrices representative of the boundary conditions at each vertex. The main point is the
development of a Maple application to test these conditions.

1. Introduction
The question addressed in this paper is that of deciding via a Maple application
whether a system of elliptic ordinary differential operators on a quantum graph
subject to given coupling conditions at the vertices is self-adjoint.
Quantum graphs are used in studying wave propagation in branching media,
electron propagation in multiply connected media, and many other problems in the
disciplines of mathematics, biology, chemistry, physics, and engineering. The survey
article [6] gives an excellent account of quantum graph models with examples from
different areas of application. Many of the problems require or assume self-adjoint
operators. An application such as ours can be used to verify self-adjointness so that
the conditions of the problem are met.
Furthermore, our application may serve as an educational tool to explore the
notion of self-adjointness of differential operators (subject to boundary conditions).
Standard methods to find solutions to partial differential equations such as separation of variables and Fourier series that are presented in undergraduate partial
differential equations classes fundamentally rely on self-adjointness of the problem.
Those methods work, after all, because of the Spectral Theorem, which (over the
reals) holds precisely for self-adjoint operators. In fact, the vast majority of spectral
theory is concerned with, and is limited to, self-adjoint operators. These operators
constitute the most important class that appears in models, and it may possibly be
surprising for a student to see how unstable this class actually is. Perturbations,
however small, in the coefficients of the operator and the boundary conditions may
turn a self-adjoint operator into a non self-adjoint one, and therefore many familiar
techniques to attack problems (such as separation of variables) become unavailable.
Quantum graphs are particularly suitable objects for students to explore the interplay between the analysis of geometric differential operators (like the Laplace or

the Dirac operator) and the topology of the underlying space. Quantum graphs in
general can have non-trivial topology, and yet they are 1-dimensional and therefore
methods from undergraduate linear algebra and differential equations are applicable. For example, to illustrate that interplay in the context of our work, consider
d
. If the graph
a graph with a single edge and on it the Dirac operator Dx = −i dx
has two vertices connected by that edge, then it is impossible to impose (local)
boundary conditions at each vertex to make Dx self-adjoint. However, if the graph
is a loop (so has only one vertex), then Dx is self-adjoint with periodic (or transmission) boundary conditions and thus represents the Dirac operator on a circle
– its spectral family is historically the first and still the principal example of a
Fourier basis. Note that the circle is topologically non-trivial, while the interval
(2-vertex graph) is homotopic to a point. This example also shows that it may be
impossible to associate self-adjoint (local) vertex conditions for an operator on a
quantum graph. Our Maple application not only tests given vertex conditions, but
is also capable of analyzing whether a system of differential operators on a quantum
graph allows self-adjoint vertex conditions at all.
While the notion of self-adjointness is conceptually not hard to grasp, it can be
a considerable computational challenge to actually verify it. For example, consider
a quantum graph with vertices where many edges meet, and impose vertex conditions that have a high degree of coupling. Verification of self-adjointness in such
a situation is computationally extremely involved. A systematic approach aided
by computer algebra software is desirable, and our Maple application provides just
that.
The paper is organized in three parts. In the first part we recall the most
important definitions and specify the setting of the problem.
The second part deals with an algebraic approach to analyze self-adjointness
that is based on symplectic linear algebra [2, 3]. We discuss results that allow us
to determine suitable algebraic conditions to analyze the problem. Furthermore,
some simple examples are presented.
The first two parts provide the necessary foundation for the Maple application
that we propose in the third part of the paper. This section includes a thorough
explanation and discussion of the implementation. The code itself is available on
the internet at
http://math.aa.psu.edu/∼summerschool
Section 3.2 provides detailed information about accessing the files and the usage of
the Maple application. The code can also be found in the appendix.
1.1. Quantum Graphs. A graph Γ is defined as a set of vertices V = {Vp } along
with a set of edges E = {ei }, with each edge ei connecting a pair of vertices. We
will consider the case only when the sets V and E contain finitely many elements.
Let |V| and |E| denote the total number of vertices and edges, respectively.
When we consider each edge as an interval, we can view a graph as a collection of
disjoint intervals where the vertices are formed by the gluing together of endpoints
of different intervals. We will consider without loss of generality the case when each
interval (edge) ei of the graph Γ is of unit length, ei = [0, 1] for i = 1, 2, . . .. The
space of measurable and square integrable functions along an edge ei is denoted by


Z
L2 (ei ) = f defined on ei
|f (x)|2 dx < ∞ .
ei

The sum of the L2 spaces of each edge creates the L2 space for the entire graph,
2

L (Γ) =

|E|
M

L2 (ei ).

i=1

The application of a differential operator Ai to each edge ei turns the graph into a
quantum graph.
Definition 1.1. A quantum graph is a graph where each edge ei is associated with
a copy of the unit interval [0, 1] and has defined upon it a differential operator Ai .
The survey [6] contains a detailed description of quantum graphs.
1.2. Maximal and Minimal Domains. Consider a linear elliptic ordinary differential operator with smooth, possibly complex-valued coefficients
n
X
dj
A=
aj (x) j ,
(1.2)
dx
j=0
where aj (x) ∈ C ∞ ([0, 1]) and an (x) 6= 0 everywhere on [0, 1]. We assume that A is
formally self-adjoint or symmetric, i.e. A = A# where A# is the formal adjoint of
A and denotes the differential operator given by
hAf, gi = hf, A# gi for f, g ∈ Cc∞ (0, 1)
under the complex inner product
Z 1
hf, gi =
f ḡ dx,

f, g ∈ L2 ([0, 1]) .

(1.3)

0

Here Cc∞ (0, 1) denotes the set of all C ∞ -smooth functions that have compact support contained in (0, 1). The operator A = Dxn for every n = 1, 2, 3, . . . is an
d
example that satisfies these conditions, where Dx = −i dx
.
We furthermore define Dmax and Dmin to be the maximal and minimal L2 domains of A as follows:

Dmax (A) := u ∈ L2 ([0, 1]) | ∃v ∈ L2 ([0, 1]) : hv, ϕi = hu, A# ϕi ∀ϕ ∈ Cc∞ (0, 1) .
(1.4)
For u ∈ Dmax (A) define Amax u = v, where v ∈ L2 ([0, 1]) satisfies the condition
in (1.4). This definition gives rise to the maximal extension Amax : Dmax (A) →
L2 ([0, 1]) of A. Note that Cc∞ (0, 1) is dense in L2 ([0, 1]), which implies that Amax
is well-defined. It is custom to drop the subscript from Amax and simply write A.
We will use both notations in the sequel and leave the subscript where we believe
that greater clarity is warranted.
The minimal L2 -domain of A can then be defined as

Dmin (A) := u ∈ Dmax (A) | hAu, vi = hu, A# vi ∀v ∈ Dmax (A# ) .
(1.5)
Let Amin = Amax Dmin (A) denote the restriction of Amax to Dmin (A).
In addition we call the operator AD = A |D : Dmax (A) ⊇ D → L2 closed, if and
only if the graph ΓAD = {(u, Au) | u ∈ D} ⊂ L2 × L2 is closed.
We remark here that Dmax (A) determines the elements u ∈ L2 ([0, 1]) such that
Au ∈ L2 ([0, 1]), while Dmin (A) is the smallest subspace of the maximal domain
that contains Cc∞ (0, 1) such that Amin : Dmin (A) → L2 ([0, 1]) is closed.
Under our present assumptions on the operator A it can be shown that the
maximal domain Dmax (A) is exactly the Sobolev space H n [0, 1] and Dmin (A) =

H0n ([0, 1]), the H n -distributions with compact support in [0, 1]. Consequently, if
the operator A is as considered in this paper, the minimal and maximal L2 -domains
are independent of A and only see its order. Moreover, the extension AD of A is
closed for all intermediate domains Dmin (A) ⊂ D ⊂ Dmax (A) between the minimal
and the maximal L2 -domains of A.
A comment on nomenclature is in order: It is custom to think of the operator
A initially as only acting on Cc∞ (0, 1). Extensions of A are then linear operators
that act on larger spaces (typically called domains), but restrict to the given action of A on Cc∞ (0, 1). The maximal extension Amax is such an extension with
domain Dmax (A) that is obtained using the L2 -inner product, and so is Amin with
domain Dmin (A). The extensions of A that are considered here all have intermediate domains Dmin (A) ⊂ D ⊂ Dmax (A). By Proposition 1.7 further below, these
are precisely the ones that are obtained from imposing boundary conditions. These
extensions are frequently referred to also as realizations of the operator A (subject
to boundary conditions).
We consider the differential operator A on a closed interval I = [a, b] ⊂ R.
Without restriction we assume [a, b] = [0, 1]. We define the left endpoint space to
be
El (A) = {ωl u + Dmin | u ∈ Dmax } ,
where ωl ∈ C ∞ (I) denotes a left cut-off function such that ωl ≡ 1 near the left
endpoint and ωl ≡ 0 near the right endpoint. The right endpoint space Er (A) is
defined analogously with a cut-off function ωr that vanishes at the left endpoint.
Then the following holds:
Dmax (A)/Dmin (A) = El (A) ⊕ Er (A)

(1.6)

We thus have the following proposition:
Proposition 1.7. The map

Tl : C ∞ ([0, 1]) 3 f 7−→ f (0), f 0 (0), f 00 (0), . . . , f (n−1) (0) ∈ Cn
extends to a continuous linear map Tl : Dmax (A) −→ Cn that vanishes on Dmin (A),
hence factors to a map
T̂l : Dmax (A)/Dmin (A) → Cn .
The map T̂l is onto, and its kernel is Er (A). Consequently, T̂l induces an isomorphism El (A) ∼
= Cn . An analogous statement holds for the right endpoint space.
The inverse map is given by
T̂l−1 : Cn → El (A),
T̂l−1 (a0 , . . . , an−1 ) = ω(x)

n−1
X
j=0

aj j
x + Dmin (A),
j!

∞

where ω is a C -function on [0, 1] such that
ω(0) = 1,

dk ω
(0) = 0, k = 1, . . . , n − 1,
dxk

dj ω
(1) = 0, j = 0, . . . , n − 1.
dxj
We refer to the monographs [1, 2, 7, 8] for background on the above.

1.3. Symplectic Form. The primary focus will be on self-adjointness of extensions
of operators. To this end, we will utilize the symplectic formalism as discussed, e.g.,
in [2]. The precise context is elaborated in greater detail in Section 2 below. At
this point, we recall some of the basics of the symplectic formalism, especially in
the context of quantum graphs.
A complex symplectic vector space is a complex vector space V equipped with a
complex symplectic form
[·, ·] : V × V → C,
(1.8)
that satisfies for all u, v, w ∈ V and λ ∈ C the following conditions:
(1) [u + v, w] = [u, w] + [v, w].
(2) [λu, v] = λ[u, v].
(3) [u, v] = −[v, u].
That is to say, the symplectic form is conjugate bilinear and skew-Hermitian.
We call the form non-degenerate if it satisfies
(4) [u, v] = 0 for all v ∈ V ⇔ u = 0.
Under our standing assumptions on the formally self-adjoint operator A on the
interval [0, 1] as stated at the beginning of Section 1.2, we obtain that the Hermitian
inner product (1.3) gives rise to the skew-Hermitian form
Z 1
Z 1
Dmax × Dmax 3 (u, v) 7→
Au · v dx −
u · Av dx = hAu, vi − hu, Avi, (1.9)
0

0

which translates into a nondegenerate complex symplectic form
[·, ·]A : V × V → C

(1.10)

where in this case V denotes the quotient space
V = Dmax (A)/Dmin (A).
The direct sum in (1.6) is in fact a symplectic orthogonal direct sum. This means
that, in addition to the sum being direct, we have [ul , ur ]A = 0 for all ul ∈ El (A)
and ur ∈ Er (A).
1.4. Symplectic Spaces on Quantum Graphs. Now, consider a finite quantum
graph Γ with the operator A = {Ai }, where each Ai is defined on the edge ei , is
formally self-adjoint, and elliptic. The quotient space Dmax (Ai )/Dmin (Ai ) associated with every operator Ai is a symplectic vector space when equipped with the
form (1.10) that is induced by (1.9).
The maximal and minimal domains of A on the graph are by definition the direct
sum of the maximal and minimal domains of each Ai .
Dmax (A) =

|E|
M
j=1

Dmax (Aj ),

Dmin (A) =

|E|
M

Dmin (Aj ).

j=1

Let V = Dmax (A)/Dmin (A), analogously to the single interval case discussed
above. V carries a symplectic form which is induced by the symplectic forms
on the quotient spaces of each edge. This symplectic form will also be denoted
by [·, ·]A . Thus, V is the symplectic orthogonal direct sum of the left and right
endpoint spaces of the operator Ai on each edge ei :

Dmax (Ai )/Dmin (Ai ) =

i=1

|E|
M

(El (Ai ) ⊕ Er (Ai )) .

(1.11)

i=1

1.5. Vertex conditions. Let A = {Ai } be an operator on the graph Γ as above,
and consider a linear map
B : Dmax (A) → Ck
for some k ∈ N that vanishes on Dmin (A). In particular, B induces a map B̂ :
V → Ck , where V = Dmax (A)/Dmin (A). With such a map B we can associate an
abstract boundary value problem (BVP) as follows:
(
Au = f ∈ L2 (Γ)
(1.12)
Find u ∈ Dmax (A) :
Bu = 0 ∈ Ck .
Then there is an obvious relation between (BVP) and some extension ADB for some
domain Dmin ⊂ DB ⊂ Dmax given by
DB = {u ∈ Dmax (A) | Bu = 0} ,
and (1.12) is equivalent to the problem
find u ∈ DB : Au = f .
In view of our ellipticity assumptions on the operators Ai and Proposition 1.7,
we conclude that B (or B̂) expresses in fact linear relations between the values
of functions (and their derivatives) that are defined on Γ at the vertices. Thus,
every abstract boundary value problem is in fact a boundary value problem in the
classical sense. In this general setup, values of functions at different vertices could
be subjected to linear relations by B̂. If this is not the case, B (or B̂) is said to
determine local vertex conditions as defined below.
Definition 1.13. For every vertex p of the graph Γ define the vertex space
M
Vp =
(endpoint spaces connected at vertex p ) .
A domain Dmin ⊂ D ⊂ Dmax for A on Γ is said to be specified by local vertex
conditions if
M
D/Dmin =
Up
p vertex

for subspaces Up ⊂ Vp .
The symplectic vector space Dmax (A)/Dmin (A) on Γ can be represented as the
symplectic orthogonal direct sum of the vertex spaces:
M
V =
Vp .
(1.14)
p vertex

Note that in case of local vertex conditions, we impose linear relations on the
collection of boundary values of the respective functions on Γ involving only the
endpoints of edges that meet at every single vertex. We will restrict our discussion
to these local conditions since the non-local case can be reduced to the local one
[6].

Copyright © SIAM
Unauthorized reproduction of this article is prohibited

79

TESTING SELF-ADJOINTNESS ON QUANTUM GRAPHS

2. Algebraic Interpretation of Self-Adjointness
We begin this section by defining the adjoint of an operator A = {Ai } on Γ. Our
standing assumption is that each Ai is elliptic and formally self-adjoint.
Definition 2.1. Consider the operator AD with domain Dmin ⊂ D ⊂ Dmax . Then
#
the adjoint of AD is defined by (AD )∗ = A#
D ∗ , the formal adjoint operator A with
∗
domain D = {v ∈ Dmax | [u, v]A = 0 for all u ∈ D}.
In particular, if A is formally self-adjoint as we assume here, i.e. A = A# , A∗D =
AD∗ holds. The operator AD is called self-adjoint, if D = D∗ .
Since we assume A to be formally self-adjoint, we obtain the following from the
definitions of Dmin and Dmax :
(Amin )∗ = Amax , (Amax )∗ = Amin .
In this paper, we will deal with the following issues:
(1) Does A admit a self-adjoint extension at all?
(2) Does A admit self-adjoint extensions subject to local vertex conditions on
Γ?
(3) How can one effectively recognize self-adjoint extensions?
2.1. Conditions for Self-Adjointness. In this section we will discover that the
existence of self-adjoint extensions AD requires the selection of a domain D where
Dmin ⊂ D ⊂ Dmax , and U = D/Dmin is a maximally isotropic subspace of the
symplectic vector space V = Dmax /Dmin . However, self-adjointness is actually even
stronger than that. Furthermore, we will find a condition which guarantees that a
maximally isotropic subspace U yields a self-adjoint extension. The arguments are
largely based on [2].
Definition 2.2. For a subspace U ⊆ V we define the symplectic orthogonal U ∗ of
U by
U ∗ = {v ∈ V |[u, v]A = 0 ∀u ∈ U }.
Moreover, U is called isotropic if
U ⊆ U ∗.
Let Ũ ⊆ V be isotropic. If U ⊆ Ũ ⇒ U = Ũ , then U is called maximally isotropic.
Theorem 2.3. Let U ⊆ V and U ∗ its symplectic orthogonal. Then the following
holds:
dim U + dim U ∗ = dim V .
(2.4)
We now observe a connection between the adjoint operator and isotropic subspaces:
Proposition 2.5. Let U = D/Dmin . If A acts with domain D, then the domain
D∗ of the adjoint of AD satisfies D∗ /Dmin = U ∗ . Therefore self-adjointness of AD
means U = U ∗ , which is equivalent to D = D∗ .
We remark here that if A allows a self-adjoint extension, it is a necessary condition for V to be even-dimensional.
We will now discuss isotropic subspaces of a general complex symplectic vector
space (W, [·, ·]) equipped with a non-degenerate symplectic form [·, ·].

Definition 2.6. Let (W, [·, ·]) be a symplectic vector space. The symplectic invariants of (W, [·, ·]) are the numbers p(W ) and q(W ) given by
p(W ) := max{dim U | U ⊆ W , and =[u, u] ≥ 0 for all u ∈ U }
q(W ) := max{dim U | U ⊆ W , and =[u, u] ≤ 0 for all u ∈ U }.
Note that whenever U is isotropic we have that
dim(U ) ≤ min{p(W ), q(W )}.
Equality holds for maximally isotropic subspaces (see e.g. [2]).
We can discover the symplectic invariants in the following manner. Let n =
dim(W ) and select a basis {v1 , . . . , vn } of W . Then we can express any u, w ∈ W
as
n
n
X
X
u=
aj vj ,
w=
bj vj .
j=1

j=1

Writing these into the symplectic form we have that


n
n
X
X
[u, w] = 
aj vj ,
bj vj  = a1 . . .
j=1

 
b̄1
 .
an H  ..  ,

j=1

b̄n

where H = ([vi , vj ]A )1≤i,j≤n . Property (3) of the symplectic form [·, ·] in (1.8)
gives us that H is a skew-Hermitian matrix, which implies H is normal and thus
diagonalizable. Since the symplectic form is assumed to be non-degenerate, we can
find an invertible matrix Q such that


iIp̃
0
QHQ∗ =
.
(2.7)
0 −iIq̃
Lemma 2.8. p̃ = p(W ) and q̃ = q(W ).
Proof. Let {w1 , . . . , wn } be the basis of W as determined by the transformation matrix Q such that ([wi , wj ]A )1≤i,j≤n is the matrix (2.7). Let U = span{w1 , . . . , wp̃ }.
Then we have that
∀u ∈ U \{0} , =[u, u] ≥ 0
(2.9)
0
0
and so p(W ) ≥ p̃. Now take U = span{wp̃+1 , . . . , wn }. Obviously dim U = q̃ and
for any subspace W̃ of W where the condition (2.9) is met for w ∈ W̃ we have that
W̃ ∩ U 0 = {0}. Thus
dim W̃ ≤ n − q̃ = p̃.
Since U is one such W̃ we have exactly that p(W ) = p̃, and we have similar proof
for the statement that q̃ = q(W ).

This allows us to state the following corollary which we will make use of later
on.
Corollary 2.10. If W = W1 ⊕ W2 is a symplectic orthogonal direct sum, then
p(W ) = p(W1 ) + p(W2 ) and q(W ) = q(W1 ) + q(W2 ).
We are thus prepared to state a theorem (see [2]) that provides a set of conditions
that can be checked to allow for a self-adjoint extension of an operator.
Theorem 2.11. A symplectic vector space W allows for subspaces U with U = U ∗
if and only if the symplectic invariants are equal.

This implies that the operator A = {Ai } on the quantum graph has a selfadjoint extension if and only if the symplectic invariants of the quotient space
V = Dmax (A)/Dmin (A) are equal. The subspaces that satisfy U = U ∗ are precisely
the maximally isotropic subspaces of V in this case. So V is necessarily evendimensional and
dim V
dim U =
.
(2.12)
2
Due to Corollary 2.10 we can make another point here for quantum graphs. If
each single-interval operator Ai has equal symplectic invariants, the operator A
on a graph admits self-adjoint extensions, but those extensions are not necessarily
ones that are subject to local vertex conditions. Local vertex conditions require the
domain D/Dmin to be a direct sum of subspaces Up (where p denotes a vertex) and
each Up has to satisfy Up = Up∗ . This gives rise to the following result.
Theorem 2.13. Self-adjoint extensions subject to local vertex conditions exist if
and only if the symplectic invariants of each vertex space are equal. The symplectic
invariants of each vertex space are the sums of the symplectic invariants of all the
endpoint spaces forming the vertex space.
In the following we will provide verifiable criteria to check whether a subspace
U of V is maximally isotropic.
We now consider V to be of even dimension 2n and dim U = n as well as
p(V ) = q(V ). This assumption on the symplectic invariants is necessarily required
to hold for the quotient space V .
Let {v1 , . . . , v2n } be a basis of V and {u1 , . . . , un } a basis of U . Since U ⊆ V ,
the basis vectors can be written as
ui =

2n
X

aij vj ,

i = 1, . . . , n,

aij ∈ C

j=1

It is enough test the condition of isotropy (Definition 2.2) for each of the basis
elements. We have for 1 ≤ i, k ≤ n




āk,1
2n
2n
X
X
 

0=[ui , uk ]A = 
aij vj ,
akl vl  = ai,1 . . . ai,2n H  ...  ,
j=1

l=1

āk,2n

A

where H = ([vi , vj ]A )1≤i,j≤2n .
We can perform a basis transformation as explained earlier in this paper such
that H is without loss of generality of the form


iIn
0
H=
.
(2.14)
0 −iIn
In this case we calculate



0

iIn
0


āk,1
 . 


0
 . 


−iIn 
. 
āk,2n

=

[ui , uk ] = (ai,1 . . . ai,2n )

=

i · (ai,1 āk,1 + · · · + ai,n āk,n − ai,n+1 āk,n+1 − · · · − ai,2n āk,2n ) .

If we write (aij )1≤i≤n,1≤j≤2n = C D with two n × n matrices C and D, this
gives us exactly the relation


C C̄ T i,k = DD̄T i,k for all i, k
⇔

CC ∗

= DD∗ .

(2.15)

So U is isotropic under this condition and indeed maximally isotropic if the
 dimension of U is the maximal possible one, which is exactly the case if C D is of full
rank (n).
In order to be able to compare with other results, we will need the representation of H with respect to a so called standard canonical symplectic basis, i.e.,
{v1 , . . . , v2n } is such that H is of the form


0
In
H=
.
−In 0
A symplectic vector space has such a basis if and only if the symplectic invariants are
equal ([2]). Thus, we are left with the following relationship between the coefficients
aij :
ai,1 āk,n+1 + · · · + ai,n āk,2n = ai,n+1 āk,1 + · · · + ai,2n āk,n .
(2.16)
Following the same method as above, we obtain in this case that
CD∗ = DC ∗ .

(2.17)

2.2. Application to Quantum Graphs. We can now utilize the result of Section 2.1 and apply it to a finite quantum graph in order to determine whether the
given set of differential operators and boundary conditions give rise to a self-adjoint
extension. Since V is representable as the symplectic orthogonal direct sum of the
vertex spaces, a domain D as determined by local vertex conditions gives rise to
a subspace U ⊆ Dmax (A)/Dmin (A) that is a symplectic orthogonal direct sum of
subspaces Up ⊂ Vp of the vertex spaces Vp .
M
U=
Up .
p vertex

Up∗

Self-adjointness means that
= Up for all p. This entails that the conditions
for self-adjointness can be checked in terms of matrices representative of the local
vertex conditions of each vertex alone, rather than the matrix representative of the
conditions on the entire graph.
To begin, let us see how the subspace Up is determined by the local vertex
conditions. Let the operators Ai be of order ni , thus each endpoint space is ni dimensional and the vertex conditions concerning the endpoints of ei can only
consist of the function values up to the (ni − 1)st derivative,
see Proposition 1.7.
P
If d endpoints are connected at a vertex p, let np = {ni |fi meets at p} represent the total dimension of this vertex space Vp . In order to discuss self-adjointness
subject to local vertex conditions, each np must necessarily be even. The local
homogeneous boundary conditions can then be represented by the linear system
Tp fp = 0, where Tp is a np /2 × np -matrix of maximal rank np /2 and fp is a vector
whose entries consist of the boundary values of each function fi at the endpoint
ki ∈ {0, 1} of the edge ei meeting at the vertex p. Enumerating the functions
1, . . . , d (thereby allowing each function to be represented at most twice since we

allow both interval endpoints to be connected to the same vertex), we have the
representation

T
fp = f1(0) (k1 ) . . . f1(n1 −1) (k1 ) . . . fd(0) (kd ) . . . fd(nd −1) (kd )
.
Thus, the subspace Up ⊂ Vp of dimension np /2 is defined by the system
Tp fp = 0.
We let T denote the matrix representative of the direct sum of each of the matrices
Tp . Relation (2.15) requires Vp to be even dimensional, so as already said we require
that np be even, and hence np /2 will be integer valued.
With this theoretical background we can now collect all conditions needed for
self-adjoint extensions and formulate this theorem:
Theorem 2.18. Let A be an elliptic formally self-adjoint operator on a quantum
graph. A choice of domain Dmin ⊂ D ⊂ Dmax for A gives rise to a self-adjoint
extension for A subject to local vertex conditions if and only if U = D/Dmin is a
direct sum of spaces Up ⊂ Vp that satisfy Up∗ = Up for each vertex p , where the
symplectic orthogonal Up∗ is taken within the symplectic vector space Vp for each p.
The key to utilizing matrix algebra to verify the conditions stated in this theorem
is a basis transformation to diagonalize the matrix representing the symplectic form
on Vp , see (2.7) and (2.14), and then checking (2.15) for the matrix representing
the vertex condition in the new basis. See Section 3 (explanation of the algorithm)
for further details.
2.3. Common Vertex Conditions. Whenever A is the one dimensional Laplacian,
d2
A = −∆= − 2 ,
dx
acting on every edge, there are three common types of local vertex conditions that
result in self-adjoint realizations of an operator A on a graph Γ. As described in
[6], they are as follows.
2.3.1. Kirchhoff Conditions. The Kirchhoff Conditions require that f is continuous
on the entire graph, meaning that at each vertex the function values are equal, and
at each vertex the sum of the outgoing derivative values (i.e. taken in a direction
away from the vertex) vanish. That is to say,

f is continuous on Γ







∀ p,

X df
(p) = 0
dxe

i∈Ep

2.3.2. δ Conditions. The δ are similar to the Kirchoff conditions in the respect f
must be continuous, but we allow the sum of derivative values to equal a multiple
of the function value at the vertex.

f is continuous on Γ







∀ p,

X df
(p) = αp f (v)
dxe

i∈Ep

2.3.3. δ 0 Conditions. The δ 0 conditions reverse the roles of the function values and
derivatives of the δ conditions.

df
(p) are independent of e
∀ p, dx


e





∀ p,

X

f (p) = αp

i∈Ep

df
(p)
dxe

3. Maple Implementation
A Maple application was written to determine if an operator A = {Ai } on
a quantum graph subject to vertex conditions is self-adjoint, or to determine if it
allows self-adjoint vertex conditions at all. The application was developed in Maple
14 and is available online at
http://math.aa.psu.edu/∼summerschool
3.1. Algorithm. By Theorem 2.18, in order to check for self-adjointness of an
operator A on a graph, we must verify the following conditions:
(1) Each Ai is elliptic and formally self-adjoint.
(2) dim U = np /2 for all vertices p. 
(3) CC ∗ = DD∗ , where T̃ = C D and T̃ is built out of T by a change of
basis, see (2.15).
The algorithm therefore begins by assembling the coefficients of the differential
operators upon each edge into a matrix C. The entry cij corresponds to the coeffidj
cient of dx
j for the operator acting on edge i. First, two validation procedures are
performed on each operator Ai . Let n be the order of the operator Ai . To ensure
ellipticity we verify that
min |cin (x)| > 0.
x∈[0,1]

To test that the operator is formally self-adjoint, we test the following equality
Ai u(x) − A#
i u(x) =

n
X
j=0

cij (x)

n
j
X
dj
j d
u(x)
−
[cij (x)u(x)] = 0.
(−1)
dxj
dxj
j=0

These are the only global procedures in the application. All other operations are
performed separately for each vertex.
Each vertex is processed in the order in which it is input.
For condition (3) the matrix representation H̃ of the symplectic form is generated for a vertex p by plugging in a basis of each endpoint space into the symplectic
form. We use the Taylor basis 1, x, x2 /2, . . . (or 1, (x − 1), (x − 1)2 /2, . . ., respectively) multiplied by (left or right) cut-off functions here. More specifically H̃ is
assembled from diagonal blocks, where each block corresponds to the symplectic
form computed for each operator Ai at the endpoint joined to p.
Then we compute the eigenvalues und eigenvectors of H̃ via Maple; so the matrix
Q1 of all eigenvectors diagonalizes H̃. To ensure that the resulting matrix is of the
form (2.14) the matrix Q2 normalizes the coefficients of the diagonal entries. Thus,
we obtain
QH̃Q∗ = H

where Q := Q2 · Q1 is an invertible matrix and H is the matrix (2.14). The algorithm checks at this point if the symplectic invariants are equal at each vertex.
The matrix Tp , as discussed in Section 2.2, is computed from the boundary
conditions input by the user. Each row of Tp corresponds to a given condition.
For each endpoint k of edge i meeting p, a block of columns is generated by the
(n −1)
coefficients of fi (k), fi0 (k), . . . , fi i (k). The blocks are then assembled to form
the matrix. This is best illustrated through an example. Suppose three endpoints,
both endpoints 0 and 1 from one interval and the endpoint 0 from a second interval,
form a vertex p with the given condition
f1 (0) − f1 (1) = 0,

f1 (1) − f2 (0) = 0,

f10 (0) − f10 (1) + f20 (0) = 0.

The 3 × 6 matrix Tp has the form
f1 (0) f10 (0) f1 (1) f1 (0) f2 (0) f20 (0)

condition 1
1
0
−1
0
0
0
condition 2  0
0
1
0
−1
0 
condition 3
0
1
0
−1
0
1


We verify that Tp has full rank, and determine the matrix N whose columns are
the basis of ker(T ).
At this point, we are prepared to compute the matrix T̃ , which is the transformation of the vertex conditions into the basis appropriate to test Theorem 2.18,
especially condition (3) as stated above, by taking

T̃ = N t Q−1 =: C D
(3.1)
where C, D are n × n-matrices. Self-adjointness is checked by verifying that
CC ∗ − DD∗ = 0.
3.2. Usage. Currently there are three files available for download:

QG.mla Maple library file, the application module
graph-sa.mw Maple workbook file, application source code
graph-examples.mw Maple workbook file, several examples of quantum graphs
The direct link to the file QG.mla is
http://math.aa.psu.edu/∼summerschool/projects/QG.mla
and likewise for the other files with QG.mla at the end replaced by the corresponding
file name.
At the very least, the QG.mla file is needed to run the application. To get started,
where “.” can be replaced with the path to QG.mla, from within Maple type:
libname := "./QG.mla", libname;
with(QuantumGraphs);
The module contains two functions:

Copyright © SIAM
Unauthorized reproduction of this article is prohibited

86

HELENE DALLMANN AND STEVEN COULTER

qgraphsa Determines whether a quantum graph coupled with a set of local
vertex conditions gives rise to a self-adjoint extension of the
operator.
qginvariants Provides the same functionality as qgraphsa but does not check
the vertex conditions.
Both functions return a boolean value, either true or false as to whether a selfadjoint extension is possible or determined by the given vertex conditions. If false
is returned, the condition and the violating vertex are returned. If multiple vertices
violate the condition, the first offending vertex is returned. The input for a graph
Γ of n edges and p vertices is a Maple list
[A, V1 , . . . , Vp ] ,
where A is the list of operators A = [A1 , . . . , An ], and Vi is a list of lists Vi = [F, BC].
Here F is a list identifying the endpoint space present at the vertex Vi and BC is
a list representing the homogeneous boundary conditions at the vertex. A sample
input is provided in Appendix B.
3.3. Examples.
3.3.1. Example 1. We first consider the three interval system pictured in Figure 1
with the one dimensional Laplacian, −∆, acting upon each edge. Our algorithm
correctly labeled the system as self-adjoint when either the Kirchhoff, δ, or δ 0 conditions were met.
When the conditions were not met, the algorithm correctly identified the system
as being not self-adjoint and identified the offending vertex causing this to be the
case. The input and output is provided in Appendix B.

'$

'$

r
V1
&%

r
V2

&%

Figure 1. −∆ on three interval graph.

3.3.2. Example 2. We next consider the system pictured in Figure 2 under the
conditions where f is continuous and the first derivatives at both endpoints of
d
d
the loop sum to 0. Note that the left loop consists of −i dx
and i dx
, the same
operator with only a sign change. If both operators were of the same sign, the
symplectic invariants would be (1, 0) and (0, 1), respectively. If this were the case,
our algorithm would report that a self-adjoint extension is not possible.

r
V1


d
i dx

d
−i dx

'$

r
V2
−∆

&%

Figure 2. Three interval graph with multiple operators.
4. Conclusions
The ultimate result of this work was the development of a Maple application
capable of validating the conditions of Theorem 2.18 for a given quantum graph.
We hope that the application will be of benefit to those with interest in the field.
Not only can our application serve as a validation step in problem solving, it can
also be used in a more stand-alone nature for studying the effect that boundary
conditions and graph structure have on self-adjointness. It would be of great use
to develop a visual front-end for the application, including means of exploring the
spectral properties for a given graph.
Acknowledgement. The authors would like to thank Thomas Krainer for giving
advice to this project in the context of the Penn State–Göttingen Summer School
2010.
Appendix A. Maple Application
This appendix details the Maple application. The files are available online at
http://math.aa.psu.edu/∼summerschool
QuantumGraphs := module ()
local ModuleLoad, A, sf, phi, w, normalizeDiag,
normalizeColumns, subscriptIndex, ns2matrix,
splitMatrix, poly2taylor, dimTotal, diffOpOrder,
checkRank, checkFormalSA, checkZero, genOpCoeffMatrix,
generateT, generateH, generateVertexH, transform,
checkLeadingCoeff;
export qgraphsa, qginvariants;
option package;
ModuleLoad := proc ()
print("Welcome to the QuantumGraphs package");
print("Authors: Steven Coulter and Helene Dallmann,
Penn State / Gottingen Summer School 2010")
end proc;
# Generic Differential Operator
A := proc (e, u, m, lambda)
options operator, arrow;
(m(e, 1))(x)*u+add((m(e, s+1))(x)*(diff(u,‘$‘(x, s))), s = 1 .. diffOpOrder(e, m))
end proc;
# Left Symplectic Form
sf[l] := proc (e, m, u, v)

options operator, arrow;
int(A(e, expand(u), m, 0)*conjugate(expand(v)), x = 0.. 1/2)
-(int(expand(u)*conjugate(A(e, expand(v), m, 0)), x = 0 .. 1/2)) end
proc;
# Right Symplectic Form
sf[r] := proc (e, m, u, v)
options operator, arrow;
int(A(e, expand(u), m,0)*conjugate(expand(v)), x = 1/2 .. 1)
-(int(expand(u)*conjugate(A(e, expand(v),m, 0)), x = 1/2 .. 1))
end proc;
# Left Cutoff Function
phi[l] := proc (n, x)
options operator, arrow;
x^(2*n)*(1/2-x)^(2*n)
end proc;
w[l] := proc (n, x)
options operator, arrow;
(int(phi[l](n, t), t = x .. 1/2))/(int(phi[l](n, t), t = 0 .. 1/2))
end proc;
# Right Cutoff Function
phi[r] := proc (n, x)
options operator, arrow;
(1-x)^(2*n)*(x-1/2)^(2*n)
endproc;
w[r] := proc (n, x)
options operator, arrow;
(int(phi[r](n, t), t = 1/2 ..x))/(int(phi[r](n, t), t = 1/2 .. 1))
end proc;
# Normalize Diagonal Elements
normalizeDiag := proc(m)::Matrix;
local i, size, D; size := LinearAlgebra:-RowDimension(m);
D := Matrix(size);
for i to size do
D[i, i] := abs(1/sqrt(-I*m(i, i)))
end do;
return D
end proc;
# Normalize Columns
normalizeColumns := proc (m)::Matrix;
local i, size; size := LinearAlgebra:-RowDimension(m);
for i to size do
m[1 .. size, i] := LinearAlgebra:-Normalize(LinearAlgebra:-Column(m, i), 2)
end do
end proc;
# Get SubScript Index
subscriptIndex := proc (list, j)::integer;
return op(1, op(0, list[j]))
end proc;
# Join Nullspace Vectors into Matrix
ns2matrix := proc (m)::Matrix;
local NS, N, j;
NS := LinearAlgebra:-NullSpace(m);
N := NS[1];
for j from 2 to nops(NS) do
N :=n‘<,>‘(‘<|>‘(N, NS[j]))
end do;
return N
end proc;
# Vertically Split Matrix
splitMatrix := proc(m)::Matrix;
local c, E, F;
c := LinearAlgebra:-ColumnDimension(m);

E := m[1 ..LinearAlgebra:-RowDimension(m), 1 .. (1/2)*c];
F := m[1 ..LinearAlgebra:-RowDimension(m), (1/2)*c+1 .. c];
return E, F
end proc;
# Find Total Dimension
dimTotal:= proc ()::integer;
local i, d; d := 0;
for i to numIntervals do
d := d+n[i]
end do;
return d
end proc;
# Order of Differential Operator
diffOpOrder := proc (e, m)
local opp, ord, i;
opp :=convert(m(e, () .. ()), list);
ord := 0;
for i to nops(opp) do
if opp[i] <> 0 then
ord := ord+1
end if
end do;
return ord-1
end proc;
# Full Rank Test
checkRank := proc (m)
return testeq(LinearAlgebra:-Rank(m),
min(LinearAlgebra:-RowDimension(m),LinearAlgebra:-ColumnDimension(m)))
end proc;
# Formally Self Adjoint Test
checkFormalSA := proc (e, m)
local n, adj, i, j;
assume(x, ’real’);
n := nops(LinearAlgebra[Row](m, e));
adj :=(m(e, 1))(x)*u(x);
for i from 2 to n do
j := i-1;
adj := adj+(-1)^j*(diff(conjugate((m(e, i))(x))*u(x), [‘$‘(x, j)]))
end do;
return testeq(adj, A(e, u(x), m, 0))
end proc;
# Ellipticity Test
checkLeadingCoeff := proc (e, m)
local n, i, lead, value;
n := nops(LinearAlgebra[Row](m, e));
for i from n by -1 to 1 do
if m(e, n) <> 0 then
lead := (m(e, n))(x)
end if
end do;
value := evalf(minimize(abs(lead), x = 0 .. 1));
return ‘not‘(testeq(value, 0))
end proc;
# Matrix Zero Check
checkZero := proc (m)
return verify(LinearAlgebra:-ZeroMatrix(LinearAlgebra:-RowDimension(m)), m, Matrix)
end proc;
# Matrix of DiffOp Coeffs
genOpCoeffMatrix := proc (list)::Matrix;
local i, j, finished, count, coeffList, nextCoeff, OpCoeffMatrix, adjust, inc;
for i to nops(list) do
finished := 0;
j := 1;
count := nops([coeffs(expand(list[i]))]);
coeffList := [unapply(coeftayl(list[i], f = 0,0), x)];

inc := nops([coeffs(coeftayl(list[i], f = 0, 0))]);
if coeffList[1] <> (proc (x) options operator, arrow; 0 end proc) then
finished := finished+inc
end if;
while finished <> count do
nextCoeff := coeff(list[i], ((D@@j)(f))(x));
coeffList := [op(coeffList), unapply(nextCoeff, x)];
if nextCoeff <> 0 then
finished := finished+nops([coeffs(nextCoeff)])
end if;
j := j+1
end do;
adjust:= nops(coeffList)-LinearAlgebra:-ColumnDimension(OpCoeffMatrix);
if 0 <= adjust then
OpCoeffMatrix := ‘<|>‘(‘<,>‘(‘<,>‘(‘<|>‘
(OpCoeffMatrix,LinearAlgebra:-ZeroMatrix
(LinearAlgebra:-RowDimension(OpCoeffMatrix), adjust))),
convert(coeffList, Matrix)))
else
OpCoeffMatrix := ‘<|>‘(‘<,>‘(OpCoeffMatrix,‘<,>‘(‘<|>‘(
convert(coeffList, Matrix), Matrix(1, abs(adjust))))))
end if
end do;
return OpCoeffMatrix
end proc;
# Taylor Matrix
generateT := proc (V, m)::Matrix;
local T, j, k, n, count, numEndPointSpaces, coeffMatrixWidth;
numEndPointSpaces :=nops(V[1]);
coeffMatrixWidth := 0;
for j to numEndPointSpaces do
coeffMatrixWidth := coeffMatrixWidth+diffOpOrder(subscriptIndex(V[1], j), m)
end do;
T := Matrix(nops(V[2]), coeffMatrixWidth);
count := 1;
for j tobnumEndPointSpaces do
for k from 0 to diffOpOrder(subscriptIndex(V[1], j), m)-1 do
for n to nops(V[2]) do
T[n, count] := coeff(V[2][n], ((D@@k)(op(0,V[1][j])))(op(1, V[1][j])))
end do;
count := count+1
end do
end do;
return T
end proc;
# H Matrix
generateH := proc (e, side, m)::Matrix;
local k, j, H, n;
n := diffOpOrder(e, m);
H := Matrix(n);
for k to n do
for j to n do
if side = 0 then
H[k, j] := sf[l](e, m, w[l](n, x)*x^(k-1)/
factorial(k-1), w[l](n,x)*x^(j-1)/factorial(j-1))
else
H[k, j] := sf[r](e, m, w[r](n,x)*(x-1)^(k-1)/
factorial(k-1), w[r](n, x)*(x-1)^(j-1)/factorial(j-1))
end if
end do
end do;
return H
end proc;
# H Matrix for Entire Vertex
generateVertexH := proc (V, m)::Matrix;
local H, M, i;
H := Matrix();
for i to nops(V[1]) do
M := generateH(op(1, op(0,V[1][i])), op(V[1][i]), m);

H := ‘<,>‘(‘<|>‘(‘<,>‘(H,Matrix(LinearAlgebra:-RowDimension(M),
LinearAlgebra:-ColumnDimension(H))),‘<,>‘(
Matrix(LinearAlgebra:-RowDimension(H), LinearAlgebra:-ColumnDimension(M)),M)))
end do;
return H
end proc;
# Transformation
transform := proc (m)::Matrix;
local Q, P, M, E,p, n, size, i;
global fail;
E, Q := LinearAlgebra:-Eigenvectors(m);
size :=LinearAlgebra:-RowDimension(Q);
M := Matrix(size);
p := 0;
n := 0;
for i to size do
if 0 < Im(E(i)) then
M(1 .. size, p+1) := Q(1 .. size, i);
p := p+1
elif Im(E(i)) < 0 then
M(1 .. size, size-n) := Q(1 .. size, i);
n := n+1
end if
end do;
Q := M;
normalizeColumns(Q);
P :=bnormalizeDiag(LinearAlgebra:-HermitianTranspose(Q).m.Q);
return P, Q, p, n
end proc;
# Self-Adjointness Test
qgraphsa := proc (list)
local i, V, OCM, T, N, H, P, Q, p, q, M, E, F, X;
for i from 2 to nops(list) do
V[i-1] := list[i]
end do;
# Generate Coefficient Matrix,
# Test for Formal Self-Adjointness
OCM := genOpCoeffMatrix(list[1]);
for i to nops(list[1]) do
if checkFormalSA(i, OCM) = FAIL then
print(i-‘not formally self adjoint‘);
return false
end if
end do;
# Loop over Vertices
for i to nops(list)-1 do
T := generateT(V[i], OCM);
# Check Rank of Taylor Matrix
if checkRank(T) = false then
print(i-‘not full rank‘);
return false
end if;
# Transform
N := ns2matrix(T);
H := generateVertexH(V[i], OCM);
P, Q, p, q := transform(H);
# Check Invariants
if p-q <> 0 then
print(i-‘p not equal q‘);
return false
end if;
# Split
M := convert(LinearAlgebra:-Transpose(N).(1/P).
(1/LinearAlgebra:-HermitianTranspose(Q)), Matrix);

E, F := splitMatrix(M);
X := simplify(E.LinearAlgebra:-HermitianTranspose(E)
-F.LinearAlgebra:-HermitianTranspose(F));
# Check Condition
if checkZero(X) = false then
print(i-‘vertex condition fail‘);
return false
end if
end do;
return true
end proc;
# Same as ggraphsa, but only check up to invariants
qginvariants := proc (list)
local i, V, OCM, T, N, H, P, Q, p, q, M, E, F, X;
for i from 2 to nops(list) do
V[i-1] := list[i] end do;
OCM := genOpCoeffMatrix(list[1]);
for i to nops(list)-1 do
if checkFormalSA(i, OCM) = FAIL then
print(i-‘not formally self adjoint‘);
return false
end if;
H := generateVertexH(V[i], OCM);
P, Q, p, q := transform(H);
if p-q <> 0 then
print(i-‘p not equal q‘);
return false
end if
end do;
return true
end proc
end module
#Save Library
LibraryTools:-Save("/tmp/QG.mla")
libname := "/tmp/QG.mla", libname
savelib(QuantumGraphs)

Appendix B. Examples
This appendix details input/output for the example provided in the paper.
libname := "/tmp/QG.mla", libname;
with(QuantumGraphs);
#Example 1a: Laplace operator on a three edge, two vertex graph, meeting Kirchhoff conditions
B := [-((D@@2)(f))(x), -((D@@2)(f))(x), -((D@@2)(f))(x)];
V[1] := [[f[1](0), f[1](1), f[2](0)],
[f[1](0)-f[1](1), f[1](1)-f[2](0), (D(f[1]))(0)-(D(f[1]))(1)+(D(f[2]))(0)]];
V[2] := [[f[3](0), f[3](1), f[2](1)],
[f[3](0)-f[3](1), f[3](1)-f[2](1), (D(f[3]))(0)-(D(f[3]))(1)-(D(f[2]))(1)]];
qgraphsa([B, V[1], V[2]]);
true
#Examples 1b: Same configuration as previous examples with not meeting Kirchhoff conditions
B := [-((D@@2)(f))(x), -((D@@2)(f))(x), -((D@@2)(f))(x)];
V[1] := [[f[1](0), f[1](1), f[2](0)],
[f[1](0)-f[1](1), 2*f[1](1)-f[2](0), (D(f[1]))(0)-(D(f[1]))(1)+(D(f[2]))(0)]];
V[2] := [[f[3](0), f[3](1), f[2](1)],
[f[3](0)-f[3](1), f[3](1)-f[2](1), (D(f[3]))(0)-(D(f[3]))(1)+(D(f[2]))(1)]];
qgraphsa([B, V[1], V[2]]);
1 - vertex condition fail

false
#Examples 1c: Same configuration as previous examples, meeting conditions
B := [-((D@@2)(f))(x), -((D@@2)(f))(x), -((D@@2)(f))(x)];
V[1] := [[f[1](0), f[1](1), f[2](0)],
[f[1](0)-f[1](1), f[1](1)-f[2](0), (D(f[1]))(0)-(D(f[1]))(1)+(D(f[2]))(0)-f[1](0)]];
V[2] := [[f[3](0), f[3](1), f[2](1)],
[f[3](0)-f[3](1), f[3](1)-f[2](1), (D(f[3]))(0)-(D(f[3]))(1)-(D(f[2]))(1)-f[3](0)]];
qgraphsa([B, V[1], V[2]]);
true
#Examples 1d: Same configuration as previous examples, meeting ’ conditions
B := [-((D@@2)(f))(x), -((D@@2)(f))(x), -((D@@2)(f))(x)];
V[1] := [[f[1](0), f[1](1), f[2](0)],
[(D(f[1]))(0)-(D(f[1]))(1), (D(f[1]))(1)-(D(f[2]))(0), f[1](0)-f[1](1)+f[2](0)-(D(f[1]))(0)]];
V[2] := [[f[3](0), f[3](1), f[2](1)],
[(D(f[3]))(0)-(D(f[3]))(1), (D(f[3]))(1)-(D(f[2]))(1), f[3](0)-f[3](1)-f[2](1)-(D(f[3]))(0)]];
qgraphsa([B, V[1], V[2]]);
true
#Example 2 a: Laplace operator on the loop, d/(dx) on the remaining two edges.
B := [I*((D@@1)(f))(x), -I*((D@@1)(f))(x), -((D@@2)(f))(x)];
V[1] := [[f[1](1), f[2](1), f[3](0), f[3](1)],
[f[1](1)-f[2](1), f[3](0)-f[3](1), (D(f[3]))(0)-(D(f[3]))(1)]];
V[2] := [[f[1](0), f[2](0)], [f[1](0)-f[2](0)]];
qgraphsa([B, V[1], V[2]]);
true

